



Lab Manual  
for  
Linear Algebra  
by  
Jim Hefferon

*Cover:* my Chocolate Lab, Suzy.

# Preface

---

*WARNING! This is an incomplete draft, no doubt riddled with errors.*

This collection supplements the text *Linear Algebra*<sup>1</sup> with explorations to help students solidify and extend their understanding of the subject, using the mathematical software Sage.<sup>2</sup>

A major goal of any undergraduate Mathematics program is to move students toward a higher-level, more abstract, grasp of the subject. For instance, Calculus classes work on elaborate computations while later courses spend more effort on concepts and proofs, focussing less on the details of calculations.

The text *Linear Algebra* fits into this development process. Its presentation works to bring students to a deeper understanding, but it does so by expecting that for them at this point a good bit of calculation helps the process. Naturally it uses examples and practice problems that are small-sized and have manageable numbers: an assignment to by-hand multiply a pair of three by three matrices of small integers will build intuition, whereas asking students to do that same question with twenty by twenty matrices of ten decimal place numbers would be badgering.

However, an instructor can be concerned that this misses a chance to make the point that Linear Algebra is widely applied, or to develop students's understanding through explorations that are not hindered by the mechanics of paper and pencil. Mathematical software can mitigate these concerns by extending the reach of what is reasonable to bigger systems, harder numbers, and longer computations. For instance, an advantage of learning how to handle larger jobs is that they are more like the ones that appear when students apply Linear Algebra to other areas. Another advantage is that students see new ideas such as runtime growth measures.

Well then, why not teach straight from the computer?

Our goal is to develop a higher-level understanding of the material so we want to keep the focus on vector spaces and linear maps. Our exposition takes computation to be a tool to develop that understanding, not the main object.

Some instructors will find that their students are best served by keeping a tight focus on the core material, and leaving aside altogether the work in this manual. Other instructors have students who will benefit from the increased reach that the software provides. This manual's existence, and status as a separate book, gives teachers the freedom to make the choice that suits their class.

---

<sup>1</sup>The text's home page <http://joshua.smcvt.edu/linearalgebra> has the PDF, the ancillary materials, and the  $\LaTeX$  source.

<sup>2</sup>See <http://www.sagemath.org> for the software and documentation.

## Why Sage?

In *Open Source Mathematical Software* [Joyner and Stein, 2007]<sup>1</sup> the authors argue that for Mathematics the best way forward is to use software that is Open Source.

Suppose Jane is a well-known mathematician who announces she has proved a theorem. We probably will believe her, but she knows that she will be required to produce a proof if requested. However, suppose now Jane says a theorem is true based partly on the results of software. The closest we can reasonably hope to get to a rigorous proof (without new ideas) is the open inspection and ability to use all the computer code on which the result depends. If the program is proprietary, this is not possible. We have every right to be distrustful, not only due to a vague distrust of computers but because even the best programmers regularly make mistakes.

If one reads the proof of Jane's theorem in hopes of extending her ideas or applying them in a new context, it is limiting to not have access to the inner workings of the software on which Jane's result builds.

Professionals choose their tools by balancing many factors but this argument is persuasive. We use Sage because it is very capable so students can learn a great deal from it, and because it is Free<sup>2</sup> and Open Source.<sup>3</sup>

## This manual

This is Free. Get the latest version from <http://joshua.smcvt.edu/linearalgebra>. Also see that page for the license details and for the L<sup>A</sup>T<sub>E</sub>X source. I am glad to get feedback, especially from instructors who have class-tested the material. My contact information is on the same page.

The computer output included here was generated automatically (I have automatically edited some lines). This is my Sage.

```
1 'Sage Version 5.2, Release Date: 2012-07-25'
```

## Reading this manual

Here I don't define all the terms or prove all the results. So a person should read the material after covering the associated chapter in the book, using that for reference.

The association between chapters here and chapters in the book is roughly: *Python and Sage* is an introduction that does not depend on the book, *Gauss's Method* is for Chapter One, *Vector Spaces* is for Chapter Two, *Matrices, Maps, and Singular Value Decomposition* go with Chapter Three, *Geometry of Linear Maps* goes best with Chapter Four, and *Eigenvalues* fits with the material from Chapter Five (it mentions Jordan Form, but only relies on the material up to the traditional ending spot of Diagonalization.)

<sup>1</sup>See <http://www.ams.org/notices/200710/tx071001279p.pdf> for the full text. <sup>2</sup>The Free Software Foundation page <http://www.gnu.org/philosophy/free-sw.html> gives background and a definition. <sup>3</sup>See <http://opensource.org/osd.html> for a definition.

## Acknowledgements

I am glad for this chance to thank the Sage Development Team. In particular, without [[Sage Development Team, 2012b](#)] this work would not have happened. I am glad also for the chance to mention [[Beezer, 2011](#)] as an inspiration. Finally, I am grateful to Saint Michael's College for the time to work on this.

*We emphasize practice.*

–[Shunryu Suzuki](#) [2006]

*[A]n orderly presentation is not necessarily bad  
but by itself may be insufficient.*

–[Ron Brandt](#) [1998]

Jim Hefferon  
Mathematics, Saint Michael's College  
Colchester, Vermont USA  
2012-Sep-10

## Contents

Python and Sage . . . . .	1
Gauss's Method . . . . .	13



# Python and Sage

---

To work through the Linear Algebra in this manual you need to be acquainted with how to run Sage. Sage uses the computer language Python so we'll start with that.

## Python

Python is a popular computer language,<sup>1</sup> often used for scripting, that is appealing for its simple style and powerful libraries. The significance of 'scripting' is that Sage uses it in this way, as a glue to bring together separate parts.

Python is Free. If your operating system doesn't provide it then go to the home page [www.python.org](http://www.python.org) and follow the download and installation instructions. Also at that site is Python's excellent tutorial. That tutorial is thorough; here you will see only enough Python to get started. For a more comprehensive introduction see [Python Team \[2012b\]](#).

*Comment.* There is a new version, Python 3, with some differences. Here we stick to the older version because that is what Sage uses. The examples below were produced directly from Python and Sage when this manual was generated so they should be exactly what you see,<sup>2</sup> unless your version is quite different than mine. Here is my version of Python.

```
1 >>> import sys
2 >>> print sys.version
3 2.7.3 (default, Aug 1 2012, 05:16:07)
4 [GCC 4.6.3]
```

**Basics** Start Python, for instance by entering `python` at a command line. You'll get a couple of lines of information followed by three greater-than characters.

```
1 >>>
```

This is a prompt. It lets you experiment: if you type Python code and *<Enter>* then the system will read your code, evaluate it, and print the result. We will see below how to write and run whole programs but for now we will experiment. You can always leave the prompt with *<Ctrl>-D*.

Try entering these expressions (double star is exponentiation).

```
1 >>> 2 - (-1)
2 3
3 >>> 1 + 2*3
4 7
```

---

<sup>1</sup>Written by a fan of Monty Python. <sup>2</sup>Lines that are too long are split to fit.

```
5 >>> 2**3
6 8
```

Part of Python's appeal is that doing simple things tend to be easy. Here is how you print something to the screen.

```
1 >>> print 1, "plus", 2, "equals", 3
2 1 plus 2 equals 3
```

Often you can debug just by putting in commands to print things, and having a straightforward print operator helps with that.

As in any other computer language, variables give you a place to keep values. The first line below puts one in the place called `i` and the second line uses that.

```
1 >>> i = 1
2 >>> i + 1
3 2
```

In some programming languages you must declare the 'type' of a variable before you use it; for instance you would have to declare that `i` is an integer before you could set `i = 1`. In contrast, Python deduces the type of a variable based on what you do to it—above we assigned 1 to `i` so Python figured that it must be an integer. Further, we can change how we use the variable and Python will go along; here we change what is in `x` from an integer to a string.

```
1 >>> x = 1
2 >>> x
3 1
4 >>> x = 'a'
5 >>> x
6 'a'
```

Python lets you assign multiple values simultaneously.

```
1 >>> first_day, last_day = 0, 365
2 >>> first_day
3 0
4 >>> last_day
5 365
```

Python computes the right side, left to right, and those values are assigned to the variables on the left. We will often use this construct.

Python complains by *raising an exception* and giving an error message. For instance, we cannot combine a string and an integer.

```
1 >>> 'a'+1
2 Traceback (most recent call last):
3 File "<stdin>", line 1, in <module>
4 TypeError: cannot concatenate 'str' and 'int' objects
```

The error message's bottom line is the useful one.

Make a comment of the rest of the line with a hash mark `#`.

```
1 >>> t = 2.2
2 >>> d = (0.5) * 9.8 * (t**2) # d in meters
3 >>> d
4 23.716000000000005
```



(Comments are more useful in a program than at the prompt.) Programmers often comment an entire line by starting that line with a hash.

As in the listing above, we can get real numbers and even complex numbers.<sup>1</sup>

```
1 >>> 5.774 * 3
2 17.322
3 >>> (3+2j) - (1-4j)
4 (2+6j)
```

As engineers do, Python uses *j* for the square root of  $-1$ , not *i* as is traditional in Mathematics.

The examples above show addition, subtraction, multiplication, and exponentiation. Division has an awkward point. Python was originally designed to have the division bar `/` mean real number division when at least one of the two numbers is real. However between two integers the division bar was taken to mean a quotient, as in “2 goes into 5 with quotient 2 and remainder 1.”

```
1 >>> 5.2 / 2.0
2 2.6
3 >>> 5.2 / 2
4 2.6
5 >>> 5 / 2
6 2
```

Experience shows this was a mistake. One of the changes in Python 3 is that the quotient operation has become `//` while the single-slash operator is always real division. In the Python 2 we are using, you must make sure that at least one number in a division is real.

```
1 >>> x = 5
2 >>> y = 2
3 >>> (1.0*x) / y
4 2.5
```

Incidentally, the integer remainder operation (sometimes called ‘modulus’) uses a percent character: `5 % 2` returns 1.

Variables can also represent truth values; these are *Booleans*.

```
1 >>> yankees_stink = True
2 >>> yankees_stink
3 True
```

You need the initial capital: `True` or `False`, not `true` or `false`.

Above we saw a string consisting of text between single quotes. You can use either single quotes or double quotes, as long as you use the same at both ends of the string. Here `x` and `y` are double-quoted, which makes sense because they contain apostrophes.

```
1 >>> x = "I'm Popeye the sailor man"
2 >>> y = "I yam what I yam and that's all what I yam"
3 >>> x + ', ' + y
4 "I'm Popeye the sailor man, I yam what I yam and that's all what I yam"
```

<sup>1</sup>Of course these aren’t actually real numbers, instead they are floating point numbers, a system that models the reals and is built into your computer’s hardware. In the prior example the distinction leaks through since its bottom line ends in 00000000005, marking where the computer’s binary approximation does not perfectly match the real that you expect. Similarly Python’s integers aren’t the integers that you studied in grade school since there is a largest one (give Python the command `import sys` followed by `sys.maxint`). However, while the issue of representation is fascinating—see [Python Team \[2012a\]](#) and [Goldberg \[1991\]](#)—we shall ignore it and just call them integers and reals.

The `+` operation concatenates strings. Inside a double-quoted string use slash-*n* `\n` to get a newline.

A string marked by three sets of quotes can contain line breaks.

```

1 >>> a = """THE ROAD TO WISDOM
2 ...
3 ... The road to wisdom?
4 ... -- Well, it's plain
5 ... and simple to express:
6 ... Err
7 ... and err
8 ... and err again
9 ... but less
10 ... and less
11 ... and less. --Piet Hein"""

```

The three dots at the start of lines after the first is Python's read-eval-print prompt telling you that what you have typed is not complete. We'll see below that a common use for triple-quoted strings is as documentation.

A Python *dictionary* is a finite function. That is, it is a finite set of ordered pairs  $\langle \text{key}, \text{value} \rangle$  subject to the restriction that no key can be used twice. Dictionaries are a simple database.

```

1 >>> english_words = {'one': 1, 'two': 2, 'three': 3}
2 >>> english_words['one']
3 1
4 >>> english_words['four'] = 4
5 >>> english_words
6 {'four': 4, 'three': 3, 'two': 2, 'one': 1}

```

Don't be mislead by this example, the words do not always just come in the reverse of the order in which you entered them. A dictionary's elements will be listed in no apparently-sensible order.

If you assign to an existing key then that will replace the previous value.

```

1 >>> english_words['one'] = 5
2 >>> english_words
3 {'four': 4, 'three': 3, 'two': 2, 'one': 5}

```

Dictionaries are central to Python, in part because looking up values in a dictionary is very fast.

While dictionaries are unordered, a *list* is ordered.

```

1 >>> a = ['alpha', 'beta', 'gamma']
2 >>> b = []
3 >>> c = ['delta']
4 >>> a
5 ['alpha', 'beta', 'gamma']
6 >>> a+b+c
7 ['alpha', 'beta', 'gamma', 'delta']

```

Get an element from a list by specifying its index, its place in the list, inside square brackets. Lists are *zero-offset* indexed, that is, the initial element of the list is numbered 0. Count from the back by using negative indices.

```

1 >>> a[0]
2 'alpha'

```

```

3 >>> a[1]
4 'beta '
5 >>> a[-1]
6 'gamma '

```

Specifying two indices separated by a colon gets a *slice* of the list.

```

1 >>> a[1:3]
2 ['beta ', 'gamma ']
3 >>> a[1:2]
4 ['beta ']

```

You can add to a list.

```

1 >>> c.append('epsilon ')
2 >>> c
3 ['delta ', 'epsilon ']

```

Lists can contain anything, including other lists.

```

1 >>> x = 4
2 >>> a = ['alpha ', [True, x]]
3 >>> a
4 ['alpha ', [True, 4]]

```

The function `range` returns a list of numbers.

```

1 >>> range(10)
2 [0, 1, 2, 3, 4, 5, 6, 7, 8, 9]
3 >>> range(1,10)
4 [1, 2, 3, 4, 5, 6, 7, 8, 9]

```

By default `range` starts at 0, which is good because lists are zero-indexed. Observe also that 9 is the highest number in the list given by `range(10)`. This makes `range(10)+range(10,20)` give the same list as `range(20)`.

A *tuple* is like a list in that it is ordered.

```

1 >>> a = ('fee', 'fie', 'foe', 'fum')
2 >>> a
3 ('fee', 'fie', 'foe', 'fum')
4 >>> a[0]
5 'fee '

```

However it is unlike a list in that a tuple is not *mutable*—it cannot change.

```

1 >>> a[0] = 'phooey '
2 Traceback (most recent call last):
3 File "<stdin>", line 1, in <module>
4 TypeError: 'tuple' object does not support item assignment

```

One reason this is useful is that because of it tuples can be keys in dictionaries while list cannot.

```

1 >>> a = ['Jim', 2138]
2 >>> b = ('Jim', 2138)
3 >>> d = {a: 'active'}
4 Traceback (most recent call last):
5 File "<stdin>", line 1, in <module>

```

```

6 TypeError: unhashable type: 'list'
7 >>> d = {b: 'active'}
8 >>> d
9 {('Jim', 2138): 'active'}

```

Python has a special value `None` for when there is no sensible value for a variable. For instance, if your program keeps track of a person's address and includes a variable `apartment` then `None` is the right value for that variable when the person does not live in an apartment.

**Flow of control** Python supports the traditional ways of affecting the order of statement execution, with a twist.

```

1 >>> x = 4
2 >>> if (x == 0):
3 ...     y = 1
4 ... else:
5 ...     y = 0
6 ...
7 >>> y
8 0

```

The twist is that while many languages use braces or some other syntax to mark a block of code, Python uses indentation. (Always indent with four spaces.) Here, Python executes the single-line block `y = 1` if `x` equals 0, otherwise Python sets `y` to 0.

Notice also that double equals `==` means “is equal to.” In contrast, we have already seen that single equals is the assignment operation so that `x = 4` means “`x` is assigned the value 4.”

Python has two variants on the above `if` statement. It could have only one branch

```

1 >>> x = 4
2 >>> y = 0
3 >>> if (x == 0):
4 ...     y = 1
5 ...
6 >>> y
7 0

```

or it could have more than two branches.

```

1 >>> x = 2
2 >>> if (x == 0):
3 ...     y = 1
4 ... elif (x == 1):
5 ...     y = 0
6 ... else:
7 ...     y = -1
8 ...
9 >>> y
10 -1

```

Computers excel at iteration, looping through the same steps.

```

1 >>> for i in range(5):
2 ...     print i, "squared is", i**2
3 ...

```

```

4 0 squared is 0
5 1 squared is 1
6 2 squared is 4
7 3 squared is 9
8 4 squared is 16

```

A for loop often involves a range.

```

1 >>> x = [4, 0, 3, 0]
2 >>> for i in range(len(x)):
3 ...     if (x[i] == 0):
4 ...         print "item",i,"is zero"
5 ...     else:
6 ...         print "item",i,"is nonzero"
7 ...
8 item 0 is nonzero
9 item 1 is zero
10 item 2 is nonzero
11 item 3 is zero

```

An experienced Python person who was not trying just to illustrate `range` would instead write `for c in x:` since the `for` loop can iterate over any sequence, not just a sequence of integers.

A `for` loop is designed to execute a certain number of times. The natural way to write a loop that will run an uncertain number of times is `while`.

```

1 >>> n = 27
2 >>> i = 0
3 >>> while (n != 1):
4 ...     if (n%2 == 0):
5 ...         n = n / 2
6 ...     else:
7 ...         n = 3*n + 1
8 ...         i = i + 1
9 ...         print "i=", i
10 ...
11 i= 1
12 i= 2
13 i= 3

```

(This listing is incomplete; it takes 111 steps to finish.)<sup>1</sup> Note that “not equal” is `!=`.

The `break` command gets you out of a loop right away.

```

1 >>> for i in range(10):
2 ...     if (i == 3):
3 ...         break
4 ...     print "i is", i
5 ...
6 i is 0
7 i is 1
8 i is 2

```

A common loop construct is to run through a list performing an action on each entry. Python has a shortcut, *list comprehension*.

---

<sup>1</sup>The *Collatz conjecture* is that for any starting `n` this loop will terminate, but this is not known.

```

1 >>> a = [2**i for i in range(4)]
2 >>> a
3 [1, 2, 4, 8]
4 >>> [i-1 for i in a]
5 [0, 1, 3, 7]

```

**Functions** A *function* is a group of statements that executes when it is called, and can return values to the caller. Here is a naive version of the quadratic formula.

```

1 >>> def quad_formula(a, b, c):
2 ...     discriminant = (b**2 - 4*a*c)**(0.5)
3 ...     r1=(-1*b+discriminant) / (2.0*a)
4 ...     r2=(-1*b-discriminant) / (2.0*a)
5 ...     return (r1, r2)
6 ...
7 >>> quad_formula(1,0,-9)
8 (3.0, -3.0)
9 >>> quad_formula(1,2,1)
10 (-1.0, -1.0)

```

(One way that it is naive is that it doesn't handle complex roots gracefully.)

Functions organize code into blocks. These blocks of code may be run a number of different times, or may belong together conceptually. In a Python program most code is in functions.

At the end of the `def` line, in parentheses, are the function's *parameters*. These can take values passed in by the caller. Functions can have *optional parameters* that have a default value.

```

1 >>> def hello(name="Jim"):
2 ...     print "Hello,", name
3 ...
4 >>> hello("Fred")
5 Hello, Fred
6 >>> hello()
7 Hello, Jim

```

Sage uses this aspect of Python heavily.

Functions always return something; if a function never executes a `return` then it will return the value `None`. They can also contain multiple `return` statements.

**Objects and modules** In Mathematics, the real numbers is a set associated with some operations such as addition and multiplication. Python is *object-oriented*, which means that we can similarly bundle together data and actions (in this context the functions are called *methods*).

```

1 >>> class DatabaseRecord(object):
2 ...     def __init__(self, name, age):
3 ...         self.name = name
4 ...         self.age = age
5 ...     def salutation(self):
6 ...         print "Dear", self.name
7 ...
8 >>> a = DatabaseRecord("Jim", 53)
9 >>> a.name

```

```

10 'Jim'
11 >>> a.age
12 53
13 >>> a.salutation()
14 Dear Jim
15 >>> b = DatabaseRecord("Fred",109)
16 >>> b.salutation()
17 Dear Fred

```

This creates two *instances* of the object `DatabaseRecord`. The `class` code describes what these consist of. The above example uses the *dot notation*: to get the age data for the instance `a` you write `a.age`. (The `self` variable can be puzzling. It is how a method refers to the instance it is part of. Suppose that at the prompt you type `a.name="James"`. Then you've used the name `a` to refer to the instance so you can make the change. In contrast, inside the `class` description code there isn't any fixed instance. The `self` gives you a way to, for example, get the `name` attribute of the current instance.)

You won't be writing your own classes here but you will be using ones from the libraries of code that others have written, including the code for Sage, so you must know how to use the classes of others. For instance, Python has a library, or *module*, for math.

```

1 >>> import math
2 >>> math.pi
3 3.141592653589793
4 >>> math.factorial(5)
5 120
6 >>> math.cos(math.pi)
7 -1.0

```

The `import` statement gets the module and makes its contents available.

**Programs** The read-eval-print loop is great for small experiments but for more than four or five lines you want to put your work in a separate file and run it as a standalone program.

To write the code, use a text editor (one example is Emacs).<sup>1</sup> Use an editor with support for Python such as automatic indentation, and syntax highlighting, where the editor colors your code to make it easier to read.

Here is a first example of a Python program. Start your editor, open a file called `test.py`, and enter these lines. Note the triple-quoted documentation string at the top of the file; include such documentation in everything that you write.

```

1 # test.py
2 """test
3
4 A test program for Python.
5 """
6
7 import datetime
8
9 current = datetime.datetime.now() # get a datetime object
10 print "the month number is", current.month

```

---

<sup>1</sup>It may come with your operating system or see <http://www.gnu.org/software/emacs>.



Run it through Python (for instance, from the command line run `python test.py`) and you should see output like the month number is 9.

Next is a small game. (It uses the Python function `raw_input` that prompts the user and then collects their response.).

```

1 # guessing_game.py
2 """guessing_game
3
4 A toy game for demonstration.
5 """
6 import random
7 CHOICE = random.randint(1,10)
8
9 def test_guess(guess):
10     """Decide if the guess is correct and print a message.
11     """
12     if (guess < CHOICE):
13         print "    Sorry, your guess is too low"
14         return False
15     elif (guess > CHOICE):
16         print "    Sorry, your guess is too high"
17         return False
18     print "    You are right!"
19     return True
20
21 flag = False
22 while (not flag):
23     guess = int(raw_input("Guess an integer between 1 and 10: "))
24     flag = test_guess(guess)

```

Here is the output resulting from running this game once at the command line.

```

1 $ python guessing_game.py
2 Guess an integer between 1 and 10: 5
3     Sorry, your guess is too low
4 Guess an integer between 1 and 10: 8
5     Sorry, your guess is too high
6 Guess an integer between 1 and 10: 6
7     Sorry, your guess is too low
8 Guess an integer between 1 and 10: 7
9     You are right!

```

As earlier, note the triple-quoted documentation strings both for the file as a whole and for the function. They provide information on how to use the `guessing_game.py` program as a whole, and how to use the function inside the program. Go to the directory containing `guessing_game.py` and start the Python read-eval-print loop. At the `>>>` prompt enter `import guessing_game`. You will play through a round of the game (there is a way to avoid this but it doesn't matter here). You are using `guessing_game` as a module. Type `help("guessing_game")`. You will see the documentation, including these lines.

```

1 DESCRIPTION
2     A toy game for demonstration.
3

```

```

4 FUNCTIONS
5     test_guess(guess)
6         Decide if the guess is correct and print a message.

```

Obviously, Python got this from the file's documentation strings. In Python, and also in Sage, good practice is to always include documentation that is accessible with the `help` command. All of Sage's built-in routines do this.

## Sage

Learning what Sage can do is the goal of much of this book so this is only a very brief walk-through of preliminaries. See also [Sage Development Team, 2012a] for a more broad-based introduction.

First, if your system does not already supply it then install Sage by following the directions at [www.sagemath.org](http://www.sagemath.org).

**Command line** Sage's command line is like Python's but adapted to mathematical work. First start Sage, for instance, enter `sage` into a command line window. You get some initial text and then a prompt (leave the prompt by typing `exit` and *(Enter)*.)

```

1 sage:
    Experiment with some expressions.

1 sage: 2**3
2 8
3 sage: 2^3
4 8
5 sage: 3*1 + 4*2
6 11
7 sage: 5 == 3+3
8 False
9 sage: sin(pi/3)
10 1/2*sqrt(3)

```

The second expression shows that Sage provides a convenient shortcut for exponentiation over Python's `2**3`. The fourth expression shows that Sage sometimes returns exact results rather than an approximation. You can still get the approximation; here are three equivalent ways.

```

1 sage: sin(pi/3).numerical_approx()
2 0.866025403784439
3 sage: sin(pi/3).n()
4 0.866025403784439
5 sage: n(sin(pi/3))
6 0.866025403784439

```

The function `n()` is an abbreviation for `numerical_approx()`.

**Script** You can group Sage commands together in a file. This way you can test the commands, and also reuse them without having to retype.

Create a file with the extension `.sage`, such as `sage_normal.sage`. Enter this function and save the file.

```
1 def normal_curve(upper_limit):
2     """Approximate area under the Normal curve from 0 to upper_limit.
3     """
4     stdev = 1.0
5     mu = 0.0
6     area=numerical_integral((1/sqrt(2*pi) * e^(-0.5*(x)^2)),
7                             0, upper_limit)
8     print "area is", area[0]
```

Bring in the file with a load command.

```
1 sage: load "sage_normal.sage"
2 sage: normal_curve(1.0)
3 area is 0.341344746069
```

**Notebook** Sage also offers a browser-based interface, where you can set up worksheets to run alone or with other people, where you can easily view plots integrated with the text, and many other nice features.

From the Sage prompt run `notebook()` and work through the tutorial.

# Gauss's Method

---

Sage can solve linear systems in a number of ways. The first way is to use a general system solver, one not specialized to linear systems. The second way is to leverage the special advantages of linear systems. We'll see both.

## Systems

To enter a system of equations you must first enter single equations. So you must start with variables. We have seen one kind of variable in giving commands like these.

```
1 sage: x = 3
2 sage: 7*x
3 21
```

Here  $x$  is the name of a location to hold values. Variables in equations are something different; in the equation  $C = 2\pi \cdot r$  the two variables do not have fixed values, nor are they tied to a location in the computer's memory.

To illustrate the difference enter an unassigned variable.<sup>1</sup>

```
1 sage: y
2 -----
3                                     ----
4 NameError                           Traceback (most recent call
5                                     last)
6
7 /home/ftpmain/linear-algebra/lab/<ipython console> in <module>()
8
9 NameError: name 'y' is not defined
```

Sage defaults to expecting that  $y$  is the name of a location to hold a value. Before you use it as a symbolic variable, you must first warn the system.

```
1 sage: var('x, y')
2 (x, y)
```

<sup>1</sup>These output blocks are taken directly from Sage without any by-hand copying and pasting. This has the advantage that there is no chance of confusion caused by author-introduced differences between what is shown in this manual and what Sage actually does. But it has two drawbacks. First, Sage puts out some lines too long to fit in the block. Sometimes these lines will be split and have their tail carried to the next line, as here, while sometimes they will be truncated if the extra characters are unimportant. The second drawback is that Sage occasionally gives system-specific responses, as in the line here giving the directory. Naturally the response on your system will show your directory. This is the first time in this manual that the issue of these artifacts appears; we won't note them again.

```

3 sage: y
4 y
5 sage: 2*y
6 2*y
7 sage: k = 3
8 sage: 2*k
9 6

```

Because we haven't told Sage otherwise, it takes  $k$  as a location to hold values and it evaluates  $2*k$ . But it does not evaluate  $2*y$  because  $y$  is a symbolic variable.

With that, a system of equations is a list.

```

1 sage: var('x,y,z')
2 (x, y, z)
3 sage: eqns = [x-y+2*z == 4, 2*x+2*y == 12, x-4*z==5]

```

You must write double equals `==` in the equations instead of the assignment operator `=`. Also, you must write  $2*x$  instead of  $2x$ . Either mistake will trigger a `SyntaxError: invalid syntax`.

Solve the system with Sage's general-purpose solver.

```

1 sage: eqns = [x-y+2*z == 4, 2*x+2*y == 12, x-4*z==5]
2 sage: solve(eqns, x,y,z)
3 [[x == 5, y == 1, z == 0]]

```

You can put a parameter in the right side and solve for the variables in terms of the parameter.

```

1 sage: var('x,y,z,a')
2 (x, y, z, a)
3 sage: eqns = [x-y+2*z == a, 2*x+2*y == 12, x-4*z==5]
4 sage: solve(eqns, x,y,z)
5 [[x == 2/5*a + 17/5, y == -2/5*a + 13/5, z == 1/10*a - 2/5]]

```

**Matrices** The *solve* routine is general-purpose but for the special case of linear systems matrix notation is the best tool.

Most of the matrices in the book have entries that are fractions so here we stick to those. Sage uses 'QQ' for the rational numbers, 'RR' or 'RDF' for real numbers, 'CC' for the complex numbers, and 'ZZ' for the integers.

Enter a row of the matrix as a list, and enter the entire matrix as a list of rows.

```

1 sage: M = matrix(QQ, [[1, 2, 3], [4, 5, 6], [7, 8, 9]])
2 sage: M
3 [1 2 3]
4 [4 5 6]
5 [7 8 9]
6 sage: M[1,2]
7 6
8 sage: M.nrows()
9 3
10 sage: M.ncols()
11 3

```

Sage lists are zero-indexed, as with Python lists, so  $M[1,2]$  asks for the entry in the second row and third column.

Enter a vector in much the same way.

```
1 sage: v = vector(QQ, [2/3, -1/3, 1/2])
2 sage: v
3 (2/3, -1/3, 1/2)
4 sage: v[1]
5 -1/3
```

Sage does not worry too much about the distinction between row and column vectors. Note however that it appears with rounded brackets so that it looks different than a one-row matrix.

You can augment a matrix with a vector.

```
1 sage: M = matrix(QQ, [[1, 2, 3], [4, 5, 6], [7, 8, 9]])
2 sage: v = vector(QQ, [2/3, -1/3, 1/2])
3 sage: M_prime = M.augment(v)
4 sage: M_prime
5 [ 1 2 3 2/3]
6 [ 4 5 6 -1/3]
7 [ 7 8 9 1/2]
```

You can use an optional argument to have Sage remember the distinction between the two parts of  $M'$ .

```
1 sage: M_prime = M.augment(v, subdivide=True)
2 sage: M_prime
3 [ 1 2 3 | 2/3]
4 [ 4 5 6 | -1/3]
5 [ 7 8 9 | 1/2]
```

**Row operations** Computers are good for jobs that are tedious and error-prone. Row operations are both.

```
1 sage: M = matrix(QQ, [[0, 2, 1], [2, 0, 4], [2, -1/2, 3]])
2 sage: v = vector(QQ, [2, 1, -1/2])
3 sage: M_prime = M.augment(v, subdivide=True)
4 sage: M_prime
5 [ 0 2 1 | 2]
6 [ 2 0 4 | 1]
7 [ 2 -1/2 3 | -1/2]
```

Swap the top rows (remember that row indices start at zero).

```
1 sage: M_prime.swap_rows(0,1)
2 sage: M_prime
3 [ 2 0 4 | 1]
4 [ 0 2 1 | 2]
5 [ 2 -1/2 3 | -1/2]
```

Rescale the top row.

```
1 sage: M_prime.rescale_row(0, 1/2)
2 sage: M_prime
3 [ 1 0 2 | 1/2]
4 [ 0 2 1 | 2]
5 [ 2 -1/2 3 | -1/2]
```

Get a new bottom row by adding  $-2$  times the top to the current bottom row.

```
1 sage: M_prime.add_multiple_of_row(2,0,-2)
2 sage: M_prime
3 [ 1 0 2 | 1/2]
4 [ 0 2 1 | 2]
5 [ 0 -1/2 -1 | -3/2]
```

Finish by finding echelon form.

```
1 sage: M_prime.add_multiple_of_row(2,1,1/4)
2 sage: M_prime
3 [ 1 0 2 | 1/2]
4 [ 0 2 1 | 2]
5 [ 0 0 -3/4 | -1]
```

By the way, Sage would have given us echelon form in a single operation had we run the command `M_prime.echelon_form()`.

Now by-hand back substitution gives the solution, or we can use `solve`.

```
1 sage: var('x,y,z')
2 (x, y, z)
3 sage: eqns=[-3/4*z == -1, 2*y+z == 2, x+2*z == 1/2]
4 sage: solve(eqns, x, y, z)
5 [[x == (-13/6), y == (1/3), z == (4/3)]]
```

The operations `swap_rows`, `rescale_rows`, and `add_multiple_of_row` changed the matrix  $M'$ . Sage has related commands that return a changed matrix but leave the starting matrix unchanged.

```
1 sage: M = matrix(QQ, [[1/2, 1, -1], [1, -2, 0], [2, -1, 1]])
2 sage: v = vector(QQ, [0, 1, -2])
3 sage: M_prime = M.augment(v, subdivide=True)
4 sage: M_prime
5 [1/2 1 -1 | 0]
6 [ 1 -2 0 | 1]
7 [ 2 -1 1 | -2]
8 sage: N = M_prime.with_rescaled_row(0,2)
9 sage: M_prime
10 [1/2 1 -1 | 0]
11 [ 1 -2 0 | 1]
12 [ 2 -1 1 | -2]
13 sage: N
14 [ 1 2 -2 | 0]
15 [ 1 -2 0 | 1]
16 [ 2 -1 1 | -2]
```

Here,  $M'$  is unchanged by the routine while  $N$  is the returned changed matrix. The other two routines of this kind are `with_swapped_rows` and `with_added_multiple_of_row`.

**Nonsingular and singular systems** Resorting to `solve` after going through the row operations is artless. Sage will give reduced row echelon form straight from the augmented matrix.

```
1 sage: M = matrix(QQ, [[1/2, 1, -1], [1, -2, 0], [2, -1, 1]])
```



```

2 sage: v = vector(QQ, [0, 1, -2])
3 sage: M_prime = M.augment(v, subdivide=True)
4 sage: M_prime.rref()
5 [ 1 0 0 | -4/5]
6 [ 0 1 0 | -9/10]
7 [ 0 0 1 | -13/10]

```

In that example the matrix  $M$  on the left is nonsingular because it is square and because Gauss's Method produces echelon forms where every one of  $M$ 's columns has a leading variable. The next example starts with a different square matrix, a singular matrix, and consequently leads to echelon form systems where there are columns on the left that do not have a leading variable.

```

1 sage: M = matrix(QQ, [[1, 1, 1], [1, 2, 3], [2, 3, 4]])
2 sage: v = vector(QQ, [0, 1, 1])
3 sage: M_prime = M.augment(v, subdivide=True)
4 sage: M_prime
5 [1 1 1|0]
6 [1 2 3|1]
7 [2 3 4|1]
8 sage: M_prime.rref()
9 [ 1 0 -1|-1]
10 [ 0 1 2| 1]
11 [ 0 0 0| 0]

```

Recall that the singular case has two subcases. The first is above: because in echelon form every row that is all zeros on the left has an entry on the right that is also zero, the system has infinitely many solutions. In contrast, with the same starting matrix the example below has a row that is zeros on the left but is nonzero on the right and so the system has no solution.

```

1 sage: v = vector(QQ, [0, 1, 2])
2 sage: M_prime = M.augment(v, subdivide=True)
3 sage: M_prime.rref()
4 [ 1 0 -1| 0]
5 [ 0 1 2| 0]
6 [ 0 0 0| 1]

```

The difference between the subcases has to do with the relationships among the rows of  $M$  and the relationships among the rows of the vector. In both cases, the relationship among the rows of the matrix  $M$  is that the first two rows add to the third. In the first case the vector has the same relationship while in the second it does not.

The easy way to ensure that a zero row in the matrix on the left is associated with a zero entry in the vector on the right is to make the vector have all zeros, that is, to consider the homogeneous system associated with  $M$ .

```

1 sage: v = zero_vector(QQ, 3)
2 sage: v
3 (0, 0, 0)
4 sage: M = matrix(QQ, [[1, 1, 1], [1, 2, 3], [2, 3, 4]])
5 sage: M_prime = M.augment(v, subdivide=True)
6 sage: M_prime
7 [1 1 1|0]
8 [1 2 3|0]
9 [2 3 4|0]

```

```

10 sage: M_prime.rref()
11 [ 1  0 -1| 0]
12 [ 0  1  2| 0]
13 [ 0  0  0| 0]

```

You can get the numbers of the columns having leading entries with the `pivots` method (there is a complementary `nonpivots`).

```

1 sage: M_prime
2 [1 1 1|0]
3 [1 2 3|1]
4 [2 3 4|2]
5 sage: M_prime.pivots()
6 (0, 1, 3)
7 sage: M_prime.rref()
8 [ 1  0 -1| 0]
9 [ 0  1  2| 0]
10 [ 0  0  0| 1]

```

Because column 2 is not in the list of pivots we know that the system is singular before we see it in echelon form.

We can use this observation to write a routine that decides if a square matrix is nonsingular.

```

1 sage: def check_nonsingular(mat):
2 ....:     if not(mat.is_square()):
3 ....:         print "ERROR: mat must be square"
4 ....:         return
5 ....:     p = mat.pivots()
6 ....:     for col in range(mat.ncols()):
7 ....:         if not(col in p):
8 ....:             print "nonsingular"
9 ....:             break
10 ....:
11 sage: N = Matrix(QQ, [[1, 2, 3], [4, 5, 6], [7, 8, 9]])
12 sage: check_nonsingular(N)
13 nonsingular
14 sage: N = Matrix(QQ, [[1, 0, 0], [0, 1, 0], [0, 0, 1]])
15 sage: check_nonsingular(N)

```

Actually, Sage matrices already have a method `is_singular` but this illustrates how you can write routines to extend Sage.

**Parametrization** You can use `solve` to give the solution set of a system with infinitely many solutions. Start with the square matrix of coefficients from above, where the top two rows add to the bottom row, and adjoin a vector satisfying the same relationship to get a system with infinitely many solutions. Then convert that to a system of equations.

```

1 sage: M = matrix(QQ, [[1, 1, 1], [1, 2, 3], [2, 3, 4]])
2 sage: v = vector(QQ, [1, 0, 1])
3 sage: M_prime = M.augment(v, subdivide=True)
4 sage: M_prime
5 [1 1 1|1]
6 [1 2 3|0]

```

```

7 [2 3 4|1]
8 sage: var('x,y,z')
9 (x, y, z)
10 sage: eqns = [x+y+z == 1, x+2*y+3*z == 0, 2*x+3*y+4*z == 1]
11 sage: solve(eqns, x, y)
12 [[x == z + 2, y == -2*z - 1]]
13 sage: solve(eqns, x, y, z)
14 [[x == r1 + 2, y == -2*r1 - 1, z == r1]]

```

The first of the two `solve` calls asks Sage to solve only for  $x$  and  $y$  and so the solution is in terms of  $z$ . In the second call Sage produces a parameter of its own.

## Automation

We finish by showing two routines to automate the by-hand row reductions of the kind that the text has in the homework. These use the matrix capabilities of Sage to both describe and perform the row operations that bring a matrix to echelon form or to reduced echelon form.

**Loading and running** The source file of the script is below, at the end. First here are a few sample calls. Start Sage in the directory containing the file `gauss_method.sage`.

```

1 sage: load "gauss_method.sage"
2 sage: M = matrix(QQ, [[1/2, 1, 4], [2, 4, -1], [1, 2, 0]])
3 sage: v = vector(QQ, [-2, 5, 4])
4 sage: M_prime = M.augment(v, subdivide=True)
5 sage: gauss_method(M_prime)
6 [1/2  1  4| -2]
7 [ 2  4 -1| 5]
8 [ 1  2  0| 4]
9 take -4 times row 1 plus row 2
10 take -2 times row 1 plus row 3
11 [1/2  1  4| -2]
12 [ 0  0 -17| 13]
13 [ 0  0 -8| 8]
14 take -8/17 times row 2 plus row 3
15 [ 1/2  1  4| -2]
16 [ 0  0 -17| 13]
17 [ 0  0  0| 32/17]

```

Because the matrix has rational number elements the operations are exact — without floating point issues.

The remaining examples skip the steps to make an augmented matrix.

```

1 sage: M1 = matrix(QQ, [[2, 0, 1, 3], [-1, 1/2, 3, 1], [0, 1, 7, 5]])
2 sage: gauss_method(M1)
3 [ 2  0  1  3]
4 [-1 1/2  3  1]
5 [ 0  1  7  5]
6 take 1/2 times row 1 plus row 2
7 [ 2  0  1  3]

```

```

8 [ 0 1/2 7/2 5/2]
9 [ 0 1 7 5]
10 take -2 times row 2 plus row 3
11 [ 2 0 1 3]
12 [ 0 1/2 7/2 5/2]
13 [ 0 0 0 0]

```

The script also has a routine to go all the way to reduced echelon form.

```

1 sage: r1 = [1, 2, 3, 4]
2 sage: r2 = [1, 2, 3, 4]
3 sage: r3 = [2, 4, -1, 5]
4 sage: r4 = [1, 2, 0, 4]
5 sage: M2 = matrix(QQ, [r1, r2, r3, r4])
6 sage: gauss_jordan(M2)
7 [ 1 2 3 4]
8 [ 1 2 3 4]
9 [ 2 4 -1 5]
10 [ 1 2 0 4]
11 take -1 times row 1 plus row 2
12 take -2 times row 1 plus row 3
13 take -1 times row 1 plus row 4
14 [ 1 2 3 4]
15 [ 0 0 0 0]
16 [ 0 0 -7 -3]
17 [ 0 0 -3 0]
18 swap row 2 with row 3
19 [ 1 2 3 4]
20 [ 0 0 -7 -3]
21 [ 0 0 0 0]
22 [ 0 0 -3 0]
23 take -3/7 times row 2 plus row 4
24 [ 1 2 3 4]
25 [ 0 0 -7 -3]
26 [ 0 0 0 0]
27 [ 0 0 0 9/7]
28 swap row 3 with row 4
29 [ 1 2 3 4]
30 [ 0 0 -7 -3]
31 [ 0 0 0 9/7]
32 [ 0 0 0 0]
33 take -1/7 times row 2
34 take 7/9 times row 3
35 [ 1 2 3 4]
36 [ 0 0 1 3/7]
37 [ 0 0 0 1]
38 [ 0 0 0 0]
39 take -4 times row 3 plus row 1
40 take -3/7 times row 3 plus row 2
41 [1 2 3 0]
42 [0 0 1 0]
43 [0 0 0 1]

```

```

44 [0 0 0 0]
45 take -3 times row 2 plus row 1
46 [1 2 0 0]
47 [0 0 1 0]
48 [0 0 0 1]
49 [0 0 0 0]

```

These are naive implementations of Gauss's Method that are just for fun. (For instance they don't handle real numbers, just rationals.) But they do illustrate the point made in this manual's Preface, since a person builds intuition by doing a reasonable number of reasonably hard Gauss's Method reductions by hand, and at that point automation can take over.

**Source of `gauss_method.sage`** *Code comment:* lines 50 and 88 are too long for this page so they end with a slash \ to make Python continue on the next line.

```

1 # code gauss_method.sage
2 # Show Gauss's method and Gauss-Jordan reduction steps.
3 # 2012-Apr-20 Jim Hefferon Public Domain.
4
5 # Naive Gaussian reduction
6 def gauss_method(M, rescale_leading_entry=False):
7     """Describe the reduction to echelon form of the given matrix of
8     rationals.
9
10    M matrix of rationals e.g., M = matrix(QQ, [[..], [..], ..])
11    rescale_leading_entry=False boolean make leading entries to 1's
12
13    Returns: None. Side effect: M is reduced. Note that this is
14    echelon form, not reduced echelon form; and that this routine
15    does not end the same way as does M.echelon_form().
16
17    """
18    num_rows=M.nrows()
19    num_cols=M.ncols()
20    print M
21
22    col = 0 # all cols before this are already done
23    for row in range(0, num_rows):
24        # ?Need to swap in a nonzero entry from below
25        while (col < num_cols
26              and M[row][col] == 0):
27            for i in M.nonzero_positions_in_column(col):
28                if i > row:
29                    print " swap row", row+1, "with row", i+1
30                    M.swap_rows(row, i)
31                    print M
32                    break
33            else:
34                col += 1
35
36    if col >= num_cols:
37        break

```

```

38
39     # Now guaranteed M[row][col] != 0
40     if (rescale_leading_entry
41         and M[row][col] != 1):
42         print " take", 1/M[row][col], "times row", row+1
43         M.rescale_row(row, 1/M[row][col])
44     print M
45     change_flag=False
46     for changed_row in range(row+1, num_rows):
47         if M[changed_row][col] != 0:
48             change_flag=True
49             factor=-1*M[changed_row][col]/M[row][col]
50             print " take", factor, "times row", row+1, \
51                 "plus row", changed_row+1
52             M.add_multiple_of_row(changed_row, row, factor)
53     if change_flag:
54         print M
55     col +=1
56
57 # Naive Gauss-Jordan reduction
58 def gauss_jordan(M):
59     """Describe the reduction to reduced echelon form of the
60     given matrix of rationals.
61
62     M matrix of rationals    e.g., M = matrix(QQ, [[..], [..], ..])
63
64     Returns: None.    Side effect: M is reduced.
65
66     """
67     gauss_method(M, rescale_leading_entry=False)
68     # Get list of leading entries [le in row 0, le in row1, ..]
69     pivot_list=M.pivots()
70     # Rescale leading entries
71     change_flag=False
72     for row in range(0, len(pivot_list)):
73         col=pivot_list[row]
74         if M[row][col] != 1:
75             change_flag=True
76             print " take", 1/M[row][col], "times row", row+1
77             M.rescale_row(row, 1/M[row][col])
78     if change_flag:
79         print M
80     # Pivot
81     for row in range(len(pivot_list)-1, -1, -1):
82         col=pivot_list[row]
83         change_flag=False
84         for changed_row in range(0, row):
85             if M[changed_row, col] != 0:
86                 change_flag=True
87                 factor=-1*M[changed_row][col]/M[row][col]
88                 print " take", factor, "times row", \

```

```
89         row+1, "plus row", changed_row+1
90         M.add_multiple_of_row(changed_row, row, factor)
91     if change_flag:
92         print M
```





# Bibliography

---

- Robert A. Beezer. Sage for Linear Algebra. <http://linear.ups.edu/download/fcla-2.22-sage-4.7.1-preview.pdf>, 2011.
- David Goldberg. What every computer scientist should know about floating-point arithmetic. *ACM Computing Surveys*, 23(1):5–48, March 1991.
- Jim Hefferon. Linear Algebra. <http://joshua.smcvt.edu/linearalgebra>, 2012.
- David Joyner and William Stein. Open source mathematical software. *Notices of the AMS*, page 1279, November 2007.
- Python Team. Floating point arithmetic: issues and limitations, 2012a. URL <http://docs.python.org/2/tutorial/float.html>. [Online; accessed 17-Dec-2012].
- Python Team. The python tutorial, 2012b. URL <http://docs.python.org/2/tutorial/index.html>. [Online; accessed 17-Dec-2012].
- Ron Brandt. *Powerful Learning*. Association for Supervision and Curriculum Development, 1998.
- Sage Development Team. Sage tutorial 5.3. <http://www.sagemath.org/pdf/SageTutorial.pdf>, 2012a.
- Sage Development Team. Sage reference manual 5.3. <http://www.sagemath.org/pdf/reference.pdf>, 2012b.
- Shunryu Suzuki. *Zen Mind, Beginners Mind*. Shambhala, 2006.