

Cover: my Chocolate Lab, Suzy.

Contents

thon and Sage	. 1
auss's Method	. 13
ector Spaces	. 25
atrices	. 33
aps	. 43

Page 32 Contents

Matrices

Matrix operations are mechanical, and are therefore perfectly suited for mechanizing.

Defining

To define a matrix you can use real number entries, or complex entries, or entries from other number systems such as the rationals.

```
sage: A = matrix(RDF, [[1, 2], [3, 4]])
sage: A
[1.0 2.0]
4 [3.0 4.0]
sage: i = CDF(i)
sage: A = matrix(CDF, [[1+2*i, 3+4*i], [5+6*i, 7+8*i]])
sage: A
[1.0 + 2.0*I 3.0 + 4.0*I]
[5.0 + 6.0*I 7.0 + 8.0*I]
sage: A = matrix(QQ, [[1, 2], [3, 4]])
sage: A
[1 2]
[3 4]
```

As with the rea numbers, Sage has two choices for models of the complex numbers and we use here the one connected with the computer's hardware. By default Sage uses i for the square root of -1 (in contrast with Python, which uses j). Note that before working with complex numbers we reset i because that letter is used for many things, so resetting is a good habit.

Unless we have a reason to do otherwise, in this chapter we'll use rational numbers because the matrices easier to read — 1 is easier than 1.0 — and because the matrices in the book have rational entries.

The matrix constructor allows you to specify the number of rows and columns.

```
1 sage: B = matrix(QQ, 2, 3, [[1, 1, 1], [2, 2, 2]])
2 sage: B
3 [1 1 1]
4 [2 2 2]
```

If your specified size doesn't match the entries

```
sage:
```

then Sage's error says Number of rows does not match up with specified number. Until now we've let Sage figure out the matrix's number of rows and columns size from the entries but

Page 34 Chapter: Matrices

a shortcut to get the zero matrix is to put the number zero in the place of the entries, and there you must say which size you want.

```
1 sage: B = matrix(QQ, 2, 3, 0)
2 sage: B
3 [0 0 0]
4 [0 0 0]
```

Another place where specifying the size is a convienence is Sage's shortcut to get an identity matrix

```
1 sage: B = matrix(QQ, 2, 2, 1)
2 sage: B
3 [1 0]
4 [0 1]
```

The difference between this shortcut and the prior one is that matrix(QQ, 3, 2, 1) gives an error because an identity matrix must be square. Sage has another shortcut that can't lead to this error.

```
sage: I = identity_matrix(4)
sage: I
[1 0 0 0]
[0 1 0 0]
[0 0 1 0]
[0 0 0 1]
```

Sage has a wealth of methods on matrices. For instance, you can transpose the rows to columns or test if the matrix is *symmetric*, unchanged by transposition.

```
sage: A = matrix(QQ, [[1, 2], [3, 4]])
sage: A.transpose()
[1 3]
[2 4]
sage: A.is_symmetric()
False
```

Linear combinations

Addition and subtraction are natural.

```
1 sage: A = matrix(QQ, [[1, 2], [3, 4]])
2 sage: B = matrix(QQ, [[1, 1], [2, -2]])
3 sage: A+B
4 [2 3]
5 [5 2]
6 sage: A-B
7 [0 1]
8 [1 6]
9 sage: B-A
10 [ 0 -1]
11 [-1 -6]
```

Sage knows that adding matrices with different sizes is undefined; this gives an error.

Section: Multiplication Page 35

Some of the lins in that output block don't contain key information so the've just been truncated. But the last line (which is so long it had to be broken and wrapped twice) is where the action is. It says that the + operand is not defined between a 2×2 matrix and a 2×3 matrix.

Scalar multiplication is also natural so you have linear combinations.

```
1 sage: 3*A
2 [ 3 6]
3 [ 9 12]
4 sage: 3*A-4*B
5 [-1 2]
6 [ 1 20]
```

Multiplication

Matrix-vector product Matrix-vector multiplication is just what you would guess.

```
1 sage: A = matrix(QQ, [[1, 3, 5, 9], [0, 2, 4, 6]])
2 sage: v = vector(QQ, [1, 2, 3, 4])
3 sage: A*v
4 (58, 40)
```

The 2×4 matrix A multiplies the 4×1 column vector \vec{v} , with the vector on the right side, as $A\vec{v}$. If you try this vector on the left as v*A then Sage gives a mismatched-sizes error.

Page 36 Chapter: Matrices

```
TypeError: unsupported operand parent(s) for '*': 'Vector space of dimension 4 over Rational Field' and 'Full MatrixSpace of 2 by 4 dense matrices over Rational Field'
```

As in the earlier error traceback some lines are truncated and the final line is wrapped twice to show it all. That final line says, in short, that the product of a size 4 vector with a size 2×4 matrix is not defined.

Of course you can multiply from the left by a vector if it has a size that matches the matrix.

```
1 sage: w = vector(QQ, [3, 5])
2 sage: w*A
3 (3, 19, 35, 57)
```

In practice you will see matrix-vector multiplications done with vectors on the left, and others on the right. In the book we multiply on the right, just to settle on one rather than switch around arbitrarily (and to have the matrix application $H\vec{x}$ fit with the map application h(x)). Sage will do either, although in some ways it has a preference for the vector on the left (as we will see in Chapter 5).

Matrix-matrix product If the sizes match then Sage will multiply the matrices. Here is the product of a 2×2 matrix A and a 2×3 matrix B.

```
1 sage: A = matrix(QQ, [[2, 1], [4, 3]])
2 sage: B = matrix(QQ, [[5, 6, 7], [8, 9, 10]])
3 sage: A*B
4 [18 21 24]
5 [44 51 58]
```

Trying B*A gives an error starting TypeError: unsupportedoperand parent(s) for '*', meaning that the product operation in this order is undefined.

Same-sized square matrices have the product defined in either order.

```
1 sage: A = matrix(QQ, [[1, 2], [3, 4]])
2 sage: B = matrix(QQ, [[4, 5], [6, 7]])
3 sage: A*B
4 [16 19]
5 [36 43]
6 sage: B*A
7 [19 28]
8 [27 40]
```

They are different; matrix multiplication is not commutative.

```
1 sage: A*B == B*A
2 False
```

In fact, matrix multiplication is very noncommutative in that if you produce two $n \times n$ matrices at random then they almost surely don't commute. Sage lets us produce matrices at random.

```
1 sage: random_matrix(RDF, 3, min=-1, max=1)
2 [  0.839578752373     -0.513434560745     -0.0103952682343]
3 [  -0.650255834117     0.0742614255301     -0.535124891715]
4 [     0.244339416814     -0.875227855986     -0.272810898333]
5 sage: random_matrix(RDF, 3, min=-1, max=1)
```

Section: Multiplication Page 37

(Note the RDF. We prefer real number entries here because random_matrix is more straightforward in this case than in the rational entry case.)

Inverse Recall that if A is nonsingular then its *inverse* A^{-1} is the matrix such that $AA^{-1} = A^{-1}A$ is the identity matrix.

```
1 sage: A = matrix(QQ, [[1, 3, 1], [2, 1, 0], [4, -1, 0]])
2 sage: A.is_singular()
3 False
```

We have a formula for 2×2 matrix inverses but in the book to by-hand compute inverses for larger matrices we write the original matrix next to the identity, and then do Gauss-Jordan reduction.

```
sage: I = identity_matrix(3)
2 sage: B = A.augment(I, subdivide=True)
3 sage: B
4 [ 1 3 1 | 1 0 0]
5 [ 2 1 0 | 0 1 0 ]
6 [ 4 -1 0 | 0 0 1]
7 sage: C = B.rref()
8 sage: C
                            1/6
            0
                  0 |
                        0
                                1/6]
9 [ 1
10
      0
            1
                  0 |
                        0
                            2/3
                                -1/3]
11 [ 0
                  1 |
                        1 -13/6
                                  5/6]
```

The inverse is the resulting matrix on the right.

```
1 sage: A_inv = C.matrix_from_columns([3, 4, 5])
2 sage: A_inv
3 [ 0 1/6
               1/6]
     0 2/3 -1/3]
4
5 [ 1 -13/6
                5/6]
6 sage: A*A_inv
7 [1 0 0]
8 [0 1 0]
9 [0 0 1]
10 sage: A_inv*A
11 [1 0 0]
12 [0 1 0]
13 [0 0 1]
```

Page 38 Chapter: Matrices

Since this is an operation that Sage users do all the time, there is a standalone command.

```
1 sage: A_inv = A.inverse()
2 sage: A_inv
3 [     0     1/6     1/6]
4 [     0     2/3     -1/3]
5 [     1 -13/6     5/6]
```

One reason for finding the inverse is to make solving linear systems easier. These three systems

$$x + 3y + z = 4$$
 $x + 3y + z = 2$ $x + 3y + z = 1/2$
 $2x + y = 4$ $2x + y = -1$ $2x + y = 0$
 $4x - y = 4$ $4x - y = 5$ $4x - y = 12$

share the matrix of coefficients but have different vectors on the right side. If you have first calculated the inverse of the matrix of coefficients then solving each system takes just a matrix-vector product.

```
sage: A = matrix(QQ, [[1, 3, 1], [2, 1, 0], [4, -1, 0]])
sage: A_inv = A.inverse()
sage: v1 = vector(QQ, [4, 4, 4])
sage: v2 = vector(QQ, [2, -1, 5])
sage: v3 = vector(QQ, [1/2, 0, 12])
sage: A_inv*v1
(4/3, 4/3, -4/3)
sage: A_inv*v2
9 (2/3, -7/3, 25/3)
sage: A_inv*v3
11 (2, -4, 21/2)
```

Running time

Since computers are fast and accurate they open up the possibility of solving problems that are quite large. Large linear algebra problems occur frequently in science and engineering. In this section we will suggest what limits there are to computers. (In this section we will use matrices with real entries because they are common in applications.)

One of the limits on just how large a problem we can do is how quickly the computer program can give answers. Naturally computers take longer to perform operations on matrices that are larger but it may be that the time the program takes to compute the answer grows more quickly than does the size of the problem — for instance, when the size of the matrix doubles then the time to do the job more than doubles.

We will time the matrix inverse operation. This is an important operation; for instance, if we could do large matrix inverses quickly then we could quickly solve large linear systems, with just a matrix-vector product. We also use it because it makes a good illustration.

```
1 sage: A = matrix(RDF, [[1, 3, 1], [2, 1, 0], [4, -1, 0]])
2 sage: A
3 [ 1.0     3.0     1.0]
4 [ 2.0     1.0     0.0]
5 [ 4.0     -1.0     0.0]
```

Section: Running time Page 39

```
6 sage: A.is_singular()
7 False
8 sage: timeit('A.inverse()')
9 5 loops, best of 3: 208 μs per loop
```

Note that Sage's timeit runs the command many times and makes a best guess about how long the operation ideally takes, because on any one time your computer may have been slowed down by a disk write or some other interruption.

The inverse operation took on the order of hundreds of microseconds. A single microsecond is 0.000 001 seconds. That's fast, but then A is only a 3×3 matrix.

Worse, A is a particular 3×3 matrix, and you'd like to know how long it takes to invert a generic, or average, matrix. You could try finding the inverse of a random matrix.

```
sage: timeit('random_matrix(RDF, 3, min=-1, max=1).inverse()')
2 625 loops, best of 3: 124 μs per loop
3 sage: timeit('random_matrix(RDF, 3, min=-1, max=1).inverse()')
4 625 loops, best of 3: 123 μs per loop
5 sage: timeit('random_matrix(RDF, 3, min=-1, max=1).inverse()')
6 625 loops, best of 3: 124 μs per loop
```

Again this is of the order of hundreds of microseconds.

But this has the issue that we can't tell from it whether the time is spent generating the random matrix or finding the inverse. In addition, there is a subtler point: we also can't tell right away if this command generates many random matrices and finds each's inverse, or if it generates one random matrix and applies the inverse many times. This code at least makes that point clear. For various sizes, 3×3 , 10×10 , etc., if finds a single random matrix and then gets the time to compute its inverse.

```
1 sage: for size in [3, 10, 25, 50, 75, 100, 150, 200]:
             print "size=", size
2 . . . . .
3 . . . . 1
            M = random_matrix(RR, size, min=-1, max=1)
4 . . . . .
             timeit('M.inverse()')
5 . . . . .
6 \text{ size} = 3
7 625 loops, best of 3: 125 μs per loop
8 \text{ size} = 10
9 625 loops, best of 3: 940 μs per loop
10 \text{ size} = 25
11 25 loops, best of 3: 12 ms per loop
12 \text{ size} = 50
13 5 loops, best of 3: 92.4 ms per loop
14 \text{ size} = 75
15 5 loops, best of 3: 308 ms per loop
16 \text{ size} = 100
17 5 loops, best of 3: 727 ms per loop
18 \text{ size} = 150
19 5 loops, best of 3: 2.45 s per loop
20 \text{ size} = 200
21 5 loops, best of 3: 5.78 s per loop
```

Some of those times are in microseconds, some are in milliseconds, and some are in seconds. This table is consistently in seconds.

Page 40 Chapter: Matrices


size	seconds
3	0.000 125
10	0.000 940
25	0.012
50	0.0924
75	0.308
100	0.727
150	2.45
200	5.78

The time grows faster than the size. For instance, in going from size 25 to size 50 the time more than doubles: 0.0924/0.012 is 7.7. Similarly, increasing the size from 50 to 200 causes the time to increase by much more than a factor of four: $5.78/0.0924 \approx 62.55$.

To get a picture give Sage the data as a list of pairs.

```
sage: d = [(3, 0.000125), (10, 0.000940), (25, 0.012),
2....: (50, 0.0924), (75, 0.308), (100, 0.727),
3....: (150, 2.45), (200, 5.78)]
4 sage: g = scatter_plot(d)
5 sage: g.save("graphics/mat001.png")
```

(If you enter scatter_plot(d) at the prompt, that is, without saving it as g, then Sage will pop up a window with the graphic.)¹



The graph dramatizes that the the ratio time/size is not constant since the data clearly does not lie on a line.

Here is some more data. The times are big enough that this computer had to run overnight.

¹The graphics in this book were generated using a few more of Sage's drawing options than are shown here, both to make them look better and to better fit the page. For instance, this graphic was generated with g = scatter_plot(d, markersize=10, facecolor='#b9b9ff'). The Sage command used to get the plot here was g.save("graphics/mat001.png", figsize=[2.25,1.5], axes_pad=0.05, fontsize=7, dpi=500). We shall omit much of this code about decoration as clutter. See the Sage manual for plot options.

Section: Running time Page 41


Again the table is a neater way to present the data.

size	seconds
500	89.2
750	299.
1000	705.

Get a graph by tacking the new data onto the existing data.

```
1 sage: d = d + [(500, 89.2), (750, 299), (1000, 705)]
2 sage: g = scatter_plot(d)
3 sage: g.save("graphics/mat002.png")
```

The result is this graphic.



Note that the two graphs have different scales; if you generated this graph with the same vertical scale as the prior graph then the data would go far off the top of the page.

So a practical limit to the size of a problem that we can solve with this matrix inverse operation comes from the fact that the graph above is not a line. The time required grows much faster than the size, and just gets too large.

A major effort in Computer Science is to find fast algorithms to do various tasks. Many people have worked on tasks in Linear Algebra in particular, such as finding the inverse of a matrix, because they are so common in applications.

Page 50 Chapter: Matrices

Bibliography

Robert A. Beezer. Sage for Linear Algebra. http://linear.ups.edu/download/fcla-2.22-sage-4.7. 1-preview.pdf, 2011.

David Goldberg. What every computer scientist should know about floating-point arithmetic. *ACM Computing Surveys*, 23(1):5–48, March 1991.

Jim Hefferon. Linear Algebra. http://joshua.smcvt.edu/linearalgebra, 2012.

David Joyner and William Stein. Open source mathematical software. *Notices of the AMS*, page 1279, November 2007.

Python Team. Floating point arithmetic: issues and limitations, 2012a. URL http://docs.python.org/2/tutorial/floatingpoint.html. [Online; accessed 17-Dec-2012].

Python Team. The python tutorial, 2012b. URL http://docs.python.org/2/tutorial/index.html. [Online; accessed 17-Dec-2012].

Ron Brandt. Powerful Learning. Association for Supervision and Curriculum Development, 1998.

Sage Development Team. Sage tutorial 5.3. http://www.sagemath.org/pdf/SageTutorial.pdf, 2012a.

Sage Development Team. Sage reference manual 5.3. http://www.sagemath.org/pdf/reference.pdf, 2012b

Shunryu Suzuki. Zen Mind, Beginners Mind. Shambhala, 2006.